Appendix B

CODE

B.1 The basic equation to be solved

The basic equation to be solved is equation (**A.97**) as developed in Appendix A and reproduced below. The observed K-Coronal intensity is given by equation (**B.1**). The expressions for physical parameters in equation (**B.1**) in terms of independent variables are given in equation (**B.2**). Equation (**B.3**) gives the parameters in equation (**B.1**) for which suitable models or actual measurements need to be used.

$$
\begin{aligned}
\mathbf{I}_{\mathbf{O}}^{\mathbf{Ra}}(\lambda, \rho \mathbf{R}_{\mathbf{solar}}) = & \int_{-\infty}^{+\infty}\int_{0}^{2\pi}\int_{\cos\omega^{*}}^{1}\int_{0}^{\infty} d\lambda' \, d\cos\omega \, d\varphi \, d(x\mathbf{R}_{\mathbf{solar}}) \times \\
& \mathbf{N}_{\mathbf{e}}(r\mathbf{R}_{\mathbf{solar}}) \times \mathbf{Q}_{\mathbf{O}}^{\mathbf{Ra}}(\alpha, \Theta) \times \\
& \frac{1}{2\sqrt{\pi}\Delta b} \mathbf{I}(\lambda', \omega, x) \exp\left[-\left(\frac{\lambda - \lambda'\left(1 + \dfrac{2b^{2}\cos\omega \, w(r\mathbf{R}_{\mathbf{solar}})_{\mathbf{radial}}}{c}\right)}{2\Delta b} \right)^{2} \right]
\end{aligned}
\qquad \textbf{(B.1)}
$$

where $\mathbf{O} \equiv (//, \perp)$

// parallel to the radial plane

$\perp$ perpendicular to the radial plane

$$\begin{aligned}
&Q_{//}^{Ra} = \frac{3}{16\pi}\sigma_T\left(\sin^2\alpha + \cos^2\alpha\cos^2\Theta\right)\\
&Q_{\perp}^{Ra} = \frac{3}{16\pi}\sigma_T\left(\cos^2\alpha + \sin^2\alpha\cos^2\Theta\right)\\
&b = \cos\gamma = \cos\left(\frac{\pi-\Theta}{2}\right)\\
&\Delta = \frac{q\lambda'}{c}\\
&q = \sqrt{\frac{2kT}{m}}\\
&I(\lambda',\omega,x) = \frac{1}{\pi}\left(\frac{AU}{R_{solar}}\right)^2\left(\frac{1-u_1+u_1\cos\theta}{1-\frac{1}{3}u_1}\right)f\\
&\Theta = \pi - \cos^{-1}\left(\sin\omega\sin\varphi\sin\chi + \cos\omega\cos\chi\right)\\
&\alpha = \sin^{-1}\left(\frac{\sin\omega\sin\varphi}{\sin(\pi-\Theta)}\right)\\
&\chi = \cos^{-1}\left(\frac{x}{r}\right)\\
&\omega^* = \sin^{-1}\left(\frac{1}{r}\right)\\
&\theta = \sin^{-1}\left(r\sin\omega\right)\\
&r^2 = x^2 + \rho^2
\end{aligned} \tag{B.2}$$

$$\begin{aligned}
&u_1(\lambda') = \text{limb darkening coefficient}\\
&f(\lambda') = \text{extraterrestrial solar irradiance}\\
&N_e(rR_{solar}) = \text{electron density model}\\
&T(rR_{solar}) = \text{coronal temperature model}\\
&W(rR_{solar}) = \text{solar wind model}
\end{aligned} \tag{B.3}$$

B.2 Approximations to the $\int d\lambda'$ integral

Consider the exponent in equation (**B.1**) be represented by **y** as shown in equation (**B.4**).

$$\mathbf{y} = \frac{\lambda - \lambda'\left(1 + 2b^2 \cos\omega \frac{w}{c}\right)}{2\Delta b} \quad \text{where } \Delta = \frac{q\lambda'}{c} \tag{B.4}$$

The first approximation to be made on equation (**B.4**) is given in equation (**B.5**).

$$\Delta = \frac{q\lambda'}{c} \rightarrow \frac{q\lambda}{c} \tag{B.5}$$

This approximation is justified on the grounds that the speed of light **c** (3×10^5) **km/sec** is much larger than the thermal electron velocity **q** (**5500 km/sec**) so that a difference of about 10 Angstroms between λ and λ' would hardly have any effect on the value of Δ.

Then differentiating equation (**B.4**) with respect to λ' together with the approximation given by equation (**B.5**) gives equation (**B.6**).

$$\frac{dy}{d\lambda'} = -\frac{1 + 2b^2 \cos\omega \frac{w}{c}}{2\Delta b} \tag{B.6}$$

The second approximation to be made on the integration limits in equation (**B.4**) is given by equation (**B.7**).

$$\boxed{\begin{array}{l} \textbf{When } \lambda' \to 0 \textbf{ then } y \to \dfrac{\lambda}{2\Delta b} = \dfrac{\lambda}{2\frac{q}{c}\lambda b} = \dfrac{c}{2qb} \\ \textbf{where } c = 3\times 10^{5}\,\text{km/sec} \\ \qquad q \cong 5500\,\text{km/sec} \\ \qquad b \le 1 \\ \qquad \therefore y \to \infty \end{array}} \tag{B.7}$$

$$\textbf{When } \lambda' \to \infty \textbf{ then } y \to -\infty\left(\frac{1+2b^{2}\cos w\frac{w}{c}}{2\Delta b}\right) \to -\infty$$

Using the limits derived in equation (**B.7**), the exponent in equation (**B.1**) together with all other λ' dependent parameters reduces the integral over $\int d\lambda'$ to equation (**B.8**).

$$\boxed{\begin{array}{l} \displaystyle\int_{0}^{\infty} e^{-\left(\frac{\lambda-\lambda'\left(1+2b^{2}\cos\omega\frac{w}{c}\right)}{2\Delta b}\right)^{2}} \frac{f(\lambda')}{2\Delta b}\,d\lambda' = \frac{1}{1+2b^{2}\cos\omega\frac{w}{c}} \int_{-\infty}^{+\infty} f\left(\frac{\lambda+y(2\Delta b)}{1+2b^{2}\cos\omega\frac{w}{c}}\right) e^{-y^{2}}\,dy \\ \textbf{and } \lambda' = \dfrac{\lambda+y(2\Delta b)}{1+2b^{2}\cos\omega\frac{w}{c}} \\ \qquad = \dfrac{\lambda\left(1+2\frac{q}{c}by\right)}{1+2b^{2}\cos\omega\frac{w}{c}} \end{array}} \tag{B.8}$$

B.3 $\mathbf{b, Q_{//}^{Ra}}$ and $\mathbf{Q_{\perp}^{Ra}}$ in terms of the independent variables $\boldsymbol{\omega, \varphi}$ and x

From equation (**B.2**) the expression for **b** is given by equation (**B.9**).

$$\boxed{\begin{aligned} \mathbf{b} &= \cos\gamma \quad \text{and} \quad 2\gamma = \pi - \Theta \\ &= \cos\left(\frac{\pi - \Theta}{2}\right) \\ &= \sin\left(\frac{\Theta}{2}\right) \end{aligned}} \tag{B.9}$$

Using the trigonometric identity $\mathbf{\cos(2\gamma) = 2\cos^2(\gamma) - 1}$ on equation (**B.9**) together with expressions for $\mathbf{\cos(\pi - \Theta), \sin\chi, \cos\chi}$ **and r** from equation (**B.2**) gives the following expression for **b,** as shown in equation **(B.10)**.

$$\boxed{\begin{aligned} \mathbf{b} &= \frac{1}{\sqrt{2}}\left(1 + \cos(2\gamma)\right)^{\frac{1}{2}} \\ &= \frac{1}{\sqrt{2}}\left(1 + \cos(\pi - \Theta)\right)^{\frac{1}{2}} \\ &= \frac{1}{\sqrt{2}}\left(1 + \sin\omega \sin\chi \sin\varphi + \cos\omega \cos\chi\right)^{\frac{1}{2}} \\ &= \frac{1}{\sqrt{2}}\left(1 + \sin\omega \frac{\rho}{\sqrt{x^2 + \rho^2}} \sin\varphi + \cos\omega \frac{x}{\sqrt{x^2 + \rho^2}}\right)^{\frac{1}{2}} \\ &= \frac{1}{\sqrt{2}}\left(1 + \sqrt{1 - \cos^2\omega} \frac{\rho}{\sqrt{x^2 + \rho^2}} \sin\varphi + \cos\omega \frac{x}{\sqrt{x^2 + \rho^2}}\right)^{\frac{1}{2}} \end{aligned}} \tag{B.10}$$

From equation (**B.2**) the expression for $\mathbf{Q}_{//}^{\mathbf{Ra}}$ is given by equation (**B.11**).

$$\boxed{\begin{aligned}\mathbf{Q}_{//}^{\mathbf{Ra}} &= \frac{3}{16\pi}\sigma_T\left(\sin^2\alpha+\cos^2\alpha\cos^2\Theta\right)\\ &= \frac{3}{16\pi}\sigma_T\left(\sin^2\alpha+(1-\sin^2\alpha)\cos^2\Theta\right)\\ &= \frac{3}{16\pi}\sigma_T\left(\cos^2\Theta+\sin^2\alpha(1-\cos^2\Theta)\right)\\ &= \frac{3}{16\pi}\sigma_T\left(1-\sin^2\Theta+\sin^2\alpha(1-\cos^2\Theta)\right)\\ &= \frac{3}{16\pi}\sigma_T\left(1+\sin^2\alpha\sin^2\Theta-\sin^2\Theta\right)\end{aligned}} \tag{B.11}$$

Now, using the relationships $\sin(2\gamma)=\sin\Theta$ and $\sin\alpha=\dfrac{\sin\omega\sin\varphi}{\sin\Theta}$ from equation (**B.2**) and the trigonometric identities $\sin(2\gamma)=2\sin\gamma\cos\gamma$ and $\sin^{2\gamma}+\cos^2\gamma=1$ on equation (**B.11**) reduces equation (**B.11**) to equation (**B.12**).

$$\boxed{\mathbf{Q}_{//}^{\mathbf{Ra}}=\frac{3}{16\pi}\sigma_T\left(1+\sin^2\omega\sin^2\varphi-4\left(1-\cos^2\gamma\right)\cos^2\gamma\right)} \tag{B.12}$$

The expression for $\cos\gamma$ as a function of (ω,φ,x) is given by equation (**B.10**).

Again, from equation (**B.2**) the expression for $\mathbf{Q}_{\perp}^{\mathbf{Ra}}$ is given by equation (**B.13**).

$$\boxed{\begin{aligned} \mathbf{Q}_{\perp}^{\mathbf{Ra}} &= \frac{3}{16\pi}\sigma_T\left(\cos^2\alpha + \sin^2\alpha\cos^2\Theta\right) \\ &= \frac{3}{16\pi}\sigma_T\left(1 - \sin^2\alpha + \sin^2\alpha\cos^2\Theta\right) \\ &= \frac{3}{16\pi}\sigma_T\left(1 - \sin^2\alpha(1-\cos^2\Theta)\right) \\ &= \frac{3}{16\pi}\sigma_T\left(1 - \sin^2\alpha\sin^2\Theta\right) \\ &= \frac{3}{16\pi}\sigma_T\left(1 - \frac{\sin^2\omega\sin^2\varphi}{\sin^2\Theta}\sin^2\Theta\right) \\ &= \frac{3}{16\pi}\sigma_T\left(1 - \sin^2\omega\sin^2\varphi\right) \end{aligned}} \tag{B.13}$$

And using the relation $\sin\alpha = \dfrac{\sin\omega\sin\varphi}{\sin\Theta}$ from equation (**B.2**) in equation (**B.13**) gives equation (**B.14**).

$$\boxed{\mathbf{Q}_{\perp}^{\mathbf{Ra}} = \frac{3}{16\pi}\sigma_T\left(1 - \sin^2\omega\sin^2\varphi\right)} \tag{B.14}$$

B.4 Rewriting the basic equation with all parameters expressed in terms of the independent variables (ω, φ, x, y)

With change of variables introduced in section **B.2** and the relationships developed in section **B.3** the expression for the observed K-Coronal intensity is given by equation (**B.15**). The expressions for the physical parameters in equation (**B.15**) in terms of the independent variables are given in equation (**B.16**). Equation (**B.17**) gives the parameters in equation (**B.15**) for which suitable models or actual measurements need to be used in terms of the given distance ρ and observed wavelength λ.

$$
\begin{aligned}
&\mathbf{I_O^{Ra}(\lambda,\rho R_{solar}) = \int_{-\infty}^{+\infty}\int_{0}^{2\pi}\int_{\cos\omega^*}^{1}\int_{-\infty}^{+\infty} dy\, d\varphi\, d\cos\omega\, d(xR_{solar})\times} \\
&\quad \mathbf{N_e\left(\sqrt{x^2+\rho^2}\,R_{solar}\right)\times Q_O^{Ra}(\omega,\varphi,x)\times} \\
&\quad \mathbf{\frac{1}{\sqrt{\pi}} I(y,\omega,\varphi,x)\frac{\exp\left[-y^2\right]}{1+2b^2\cos\omega\frac{w}{c}}} \\
&\mathbf{\text{where } O \equiv (//,\perp)} \\
&\mathbf{//\ \text{parallel to the radial plane}} \\
&\mathbf{\perp\ \text{perpendicular to the radial plane}}
\end{aligned}
\qquad \mathbf{(B.15)}
$$

$$
\begin{aligned}
&\mathbf{b = \frac{1}{\sqrt{2}}\left(1+\sqrt{1-\cos^2\omega}\,\frac{\rho}{\sqrt{x^2+\rho^2}}\sin\varphi + \cos\omega\frac{\rho}{\sqrt{x^2+\rho^2}}\right)^{1/2}} \\
&\mathbf{Q_{//}^{Ra} = \frac{3}{16\pi}\sigma_T\left(1+\sin^2\omega\sin^2\varphi - 4\left(1-b^2\right)b^2\right)} \\
&\mathbf{Q_{\perp}^{Ra} = \frac{3}{16\pi}\sigma_T\left(1-\sin^2\omega\sin^2\varphi\right)} \\
&\mathbf{\Delta = \frac{q\lambda}{c}} \\
&\mathbf{q = \sqrt{\frac{2kT}{m}}} \\
&\mathbf{I(y,\omega,\varphi,x) = \frac{1}{\pi}\left(\frac{AU}{R_{solar}}\right)^2\left(\frac{1-u_1+u_1\cos\theta}{1-\frac{1}{3}u_1}\right)f} \\
&\mathbf{\cos\omega^* = \cos(\sin^{-1}\left(\frac{1}{\sqrt{x^2+\rho^2}}\right))} \\
&\mathbf{\cos\theta = \cos(\sin^{-1}\left(\sqrt{x^2+\rho^2}\,\sin\omega\right))}
\end{aligned}
\qquad \mathbf{(B.16)}
$$

$$
\begin{aligned}
&u_1(\xi) = \text{limb darkening coefficient} \\
&f(\xi) = \text{extraterrestrial solar irradiance} \\
&\xi = \frac{\lambda\left(1 + 2\frac{q}{c}by\right)}{1 + 2b^2 \cos\omega \frac{w}{c}} \\
&N_e\left(\sqrt{x^2 + \rho^2} R_{solar}\right) = \text{electron density model} \\
&T(\sqrt{x^2 + \rho^2} R_{solar}) = \text{coronal temperature model} \\
&W\left(\sqrt{x^2 + \rho^2} R_{solar}\right) = \text{solar wind model}
\end{aligned}
\tag{B.17}
$$

B.5 Changing the independent variable $\cos\omega$ to $\cos\theta$

Consider the integral in equation (**B.15**) as shown in equation (**B.18**).

$$
\int_{\cos\omega^*}^{1} d\cos\omega \tag{B.18}
$$

From equation (**B.2**) the expression for $\cos\omega^*$ and θ are given by equation (**B.19**) and equation (**B.20**), respectively.

$$
\begin{aligned}
\cos\omega^* &= \sqrt{1 - \sin^2\omega^*} \\
&= \sqrt{1 - \frac{1}{x^2 + \rho^2}}
\end{aligned}
\tag{B.19}
$$

$$
\begin{aligned}
&\theta = \sin^{-1}\left(\sqrt{x^2 + \rho^2}\sin\omega\right) \\
&\therefore \sin\omega = \frac{\sin\theta}{\sqrt{x^2 + \rho^2}} \\
&\qquad\quad = \sqrt{\frac{1 - \cos^2\theta}{x^2 + \rho^2}} \\
&\therefore \cos\omega = \cos\left(\sin^{-1}\left(\sqrt{\frac{1 - \cos^2\theta}{x^2 + \rho^2}}\right)\right)
\end{aligned}
\tag{B.20}
$$

Substituting $\cos\theta = \mu$ equation (**B.20**) can be written as shown in equation (**B.21**).

$$\cos\omega = \cos\left(\sin^{-1}\left(\sqrt{\frac{1-\mu^2}{x^2+\rho^2}}\right)\right) \tag{B.21}$$

Differentiating equation (**B.21**) with respect to μ gives equation (**B.22**).

$$\boxed{\begin{aligned}\frac{d\cos\omega}{d\mu} &= -\sin\left(\sin^{-1}\left(\sqrt{\frac{1-\mu^2}{x^2+\rho^2}}\right)\right)\frac{1}{\sqrt{1-\frac{1-\mu^2}{x^2+\rho^2}}}\times \\ &\quad \frac{1}{\sqrt{x^2+\rho^2}}\left(\frac{1}{2}\right)\left(1-\mu^2\right)^{-\frac{1}{2}}(-2\mu) \\ &= \frac{\mu}{\sqrt{x^2+\rho^2}\sqrt{x^2+\rho^2+\mu^2-1}}\end{aligned}} \tag{B.22}$$

And the limits on the integration $\int_{\cos\omega^*}^{1} d\cos\omega$ will change to the following form.

From equation (**B.16**) the expression for μ can be written as equation (**B.23**).

$$\boxed{\begin{aligned}\mu &= \cos\theta \\ &= \sqrt{1-\sin^2\theta} \\ &= \sqrt{1-r^2\sin^2\omega} \\ &= \sqrt{1-\left(x^2+\rho^2\right)\left(1-\cos^2\omega\right)}\end{aligned}} \tag{B.23}$$

$$\textbf{when } \cos\omega = 1 \textbf{ then } \mu = 1 \textbf{ and}$$

$$\textbf{when } \cos\omega = \cos\omega^* = \sqrt{1-\frac{1}{x^2+\rho^2}} \textbf{ then } \mu = 0$$

Incorporating the change of variable from ω to θ the basic equation (**B.15**) changes to equation (**B.24**).). The expressions for physical parameters in equation (**B.24**) in terms of independent variables are given in equation (**B.25**). Equation (**B.26**) gives the parameters in equation (**B.24**) for which suitable models or actual measurements need to be used in terms of distance ρ and observed wavelength λ.

$$\begin{aligned}
\mathbf{I}_{\mathbf{O}}^{\mathbf{Ra}}(\lambda, \rho \mathbf{R}_{\mathbf{solar}}) = & \int_{-\infty}^{+\infty}\int_{0}^{2\pi}\int_{0}^{1}\int_{-\infty}^{+\infty} \mathbf{dy}\, \mathbf{d}\varphi\, \mathbf{d}\mu\, \mathbf{d}(\mathbf{x}\mathbf{R}_{\mathbf{solar}}) \times \\
& \mathbf{N}_{\mathbf{e}}\left(\sqrt{\mathbf{x}^2 + \rho^2}\,\mathbf{R}_{\mathbf{solar}}\right) \times \mathbf{Q}_{\mathbf{O}}^{\mathbf{Ra}}(\mu, \varphi, \mathbf{x}) \times \\
& \frac{\mu}{\sqrt{\mathbf{x}^2 + \rho^2}\sqrt{\mathbf{x}^2 + \rho^2 + \mu^2 - \mathbf{1}}} \times \\
& \frac{\mathbf{1}}{\sqrt{\pi}} \mathbf{I}(\mathbf{y}, \mu, \varphi, \mathbf{x}) \frac{\exp\left[-\mathbf{y}^2\right]}{\mathbf{1} + \mathbf{2b}^2 \beta \frac{\mathbf{w}}{\mathbf{c}}}
\end{aligned} \qquad \textbf{(B.24)}$$

where $\mathbf{O} \equiv (//, \perp)$

// parallel to the radial plane

$\perp$ perpendicular to the radial plane

$$
\begin{aligned}
&\mathbf{b}=\frac{1}{\sqrt{2}}\left(1+\sqrt{1-\beta^2}\,\frac{\rho}{\sqrt{x^2+\rho^2}}\sin\varphi+\beta\frac{\rho}{\sqrt{x^2+\rho^2}}\right)^{1/2}\\
&Q_{//}^{Ra}=\frac{3}{16\pi}\sigma_T\left(1+\left(1-\beta^2\right)\sin^2\varphi-4\left(1-b^2\right)b^2\right)\\
&Q_{\perp}^{Ra}=\frac{3}{16\pi}\sigma_T\left(1-\left(1-\beta^2\right)\sin^2\varphi\right)\\
&\beta=\cos\left(\sin^{-1}\left(\sqrt{\frac{1-\mu^2}{x^2+\rho^2}}\right)\right)\\
&\Delta=\frac{q\lambda}{c}\\
&q=\sqrt{\frac{2kT}{m}}\\
&I(y,\mu,\varphi,x)=\frac{1}{\pi}\left(\frac{AU}{R_{solar}}\right)^2\left(\frac{1-u_1+u_1\mu}{1-\frac{1}{3}u_1}\right)f
\end{aligned}
\tag{B.25}
$$

$$
\begin{aligned}
&u_1(\xi)=\text{limb darkening coefficient}\\
&f(\xi)=\text{extraterrestrial solar irradiance}\\
&\xi=\frac{\lambda\left(1+2\frac{q}{c}by\right)}{1+2b^2\beta\frac{w}{c}}\\
&N_e\left(\sqrt{x^2+\rho^2}R_{solar}\right)=\text{electron density model}\\
&T(\sqrt{x^2+\rho^2}R_{solar})=\text{coronal temperature model}\\
&W\left(\sqrt{x^2+\rho^2}R_{solar}\right)=\text{solar wind model}
\end{aligned}
\tag{B.26}
$$

B.6 Expression for the extraterrestrial solar irradiance $f(\xi)$



Figure (B.1). The plot of extraterrestrial solar irradiance VS wavelength obtained with a Fourier Transform Spectrometer at the McMath/Pierce Solar Telescope on Kitt Peak, Arizona.

The spectrum of extraterrestrial solar irradiance shown in figure (**B.1**) was downloaded from ***ftp.noao.edu/fts/fluxatl.*** The spectrum was then recreated in intervals of 0.0025 Angstroms through linear interpolation. The reasons for this change will be apparent in section (**B.10**). Here $f(\xi)$ is the wavelength dependent extraterrestrial solar irradiance with wavelength ξ measured in Angstroms.

B.7 Expressions for the limb-darkening coefficient $u_1(\xi)$



Figure (B.2). The plot of limb-darkening coefficient versus wavelength with linear approximation super imposed.

Using the wavelength dependent limb-darkening coefficient from ***Astrophysical Quantities*** by ***Allen*** the following linear fits were made for the different wavelength regions.

(1) 3000 – 3200 Angstroms

$$\mathbf{u_1(\xi) = \frac{538.0}{200.0} - \frac{0.11 \times \xi}{200.0}} \qquad \textbf{(B.27)}$$

(2) 3200 – 3700 Angstroms

$$\mathbf{u_1(\xi) = 0.922 - \frac{(\xi - 3220.0) \times 0.120}{480.0}} \qquad \textbf{(B.28)}$$

(3) 3700 – 5000 Angstroms

$$\mathbf{u_1(\xi) = 0.862 - \frac{(\xi - 3700.0) \times 0.212}{1300.0}} \qquad \textbf{(B.29)}$$

(4) 5000 – 5500 Angstroms

$$\mathbf{u_1(\xi) = \frac{625.0}{500.0} - \frac{0.06 \times \xi}{500.0}} \qquad \textbf{(B.30)}$$

In the above equations $\boldsymbol{\xi}$ is the wavelength measured in Angstroms.

B.8 Expression for the coronal electron density model $\mathbf{N_e\left(\sqrt{x^2+\rho^2}\,R_{solar}\right)}$

If the eclipse takes place at sunspot maximum, the corona has an approximate circular form. This means that the coronal rays are most probably equally distributed between the polar and the equatorial regions of the Sun. Since the Sun is expected to reach sunspot maximum in the year 2000 the solar corona at the August 11, 1999 total solar eclipse will assumed to be of circular form. This assumption justifies using a radial dependent expression for the electron number density.

From the analysis of the photometric data on ten eclipses from 1905 to 1929 Baumbach (1937) deduced the following expression for the coronal electron density.

$$\mathbf{N_e(r) = 10^8\left(0.036r^{-1.5} + 1.55r^{-6} + 2.99r^{-16}\right)cm^{-3}} \quad \text{where } \mathbf{r = \sqrt{x^2+\rho^2}} \qquad \mathbf{(B.31)}$$

Cram (1976) used the expression given in equation (**B.32**).

$$\mathbf{N_e(r) = 1.67 \times 10^{(4+4.04/r)}\ cm^{-3}} \quad \text{where } \mathbf{r = \sqrt{x^2+\rho^2}} \qquad \mathbf{(B.32)}$$

Equation (**B.32**) agrees within 2% in the region $\mathbf{1.5 < r < 2.0}$ to the minimum equator model given by Van de Hulst (1950). In this code the electron density model given by equation (**B.31**) was used. Figure (**B.3**) shows the electron density plots given by equation (**B.31**) and equation (**B.32**).



Figure (B.3). The plot of electron number density versus the radial distance from the solar surface for the two models given by equation (B.31) and equation (B.32).

B.9 Expressions for the coronal temperature model $\mathbf{T}\left(\sqrt{x^2+\rho^2}\,\mathbf{R}_{\text{solar}}\right)$ and the solar wind model $\mathbf{W}\left(\sqrt{x^2+\rho^2}\,\mathbf{R}_{\text{solar}}\right)$

In this code the corona is considered to be isothermal which implies that that the temperature **T** is a constant. As regards the solar wind **W**, it is considered to be radial, isotropic and constant. The idea is to construct multiple models for different combinations of temperature **T** and solar wind velocity **W**. Nevertheless, the code does not restrict from applying a coordinate dependent temperature and wind profiles that are measured independently.

B.10 Evaluation of the integrals

Rewriting equation (**B.24**) incorporating all of the parameters introduced in sections **B.6** to **B.9** give equation (**B.33**) and (**B.24**) for $\mathbf{I}_{//}^{\mathbf{Ra}}$ and $\mathbf{I}_{\perp}^{\mathbf{Ra}}$, respectively.

$$
\begin{aligned}
&\mathbf{I}_{//}^{\mathbf{Ra}}(\lambda,\rho) = \mathbf{constant} \times \int_{-\infty}^{+\infty}\int_{0}^{2\pi}\int_{0}^{1}\int_{-\infty}^{+\infty} \mathbf{dy\, d\varphi\, d\mu\, dx} \times \\
&10^{8} \times \left(0.036 \times \left(\sqrt{x^2+\rho^2}\right)^{-1.5} + 1.55 \times \left(\sqrt{x^2+\rho^2}\right)^{-6} + 2.99 \times \left(\sqrt{x^2+\rho^2}\right)^{-16}\right) \times \\
&\frac{\mu}{\sqrt{x^2+\rho^2}\sqrt{x^2+\rho^2+\mu^2-1}} \times \\
&\left(1+\left(1-\beta^2\right)\sin^2\varphi - 4\left(1-b^2\right)b^2\right) \times \\
&\frac{1-u_1(\xi)+\mu u_1(\xi)}{1-\frac{1}{3}u_1(\xi)} f(\xi) \times \\
&\frac{\exp\left[-y^2\right]}{1+2b^2\beta\frac{w}{c}} \quad \mathbf{where} \\
&\mathbf{constant} = \frac{3\times\sigma_T (AU)^2}{16\pi^{5/2} R_{Solar}} \\
&b = \frac{1}{\sqrt{2}}\left(1+\rho\sqrt{\frac{1-\beta^2}{x^2+\rho^2}}\sin\varphi + \frac{\beta\rho}{\sqrt{x^2+\rho^2}}\right)^{1/2} \\
&\beta = \cos\left(\sin^{-1}\left(\sqrt{\frac{1-\mu^2}{x^2+\rho^2}}\right)\right) \\
&\xi = \frac{\lambda\left(1+2\frac{q}{c}by\right)}{1+2b^2\beta\frac{w}{c}} \quad \mathbf{and} \quad q = \sqrt{\frac{2kT}{m}}
\end{aligned}
\qquad \mathbf{(B.33)}
$$

$$I_{\perp}^{Ra}(\lambda,\rho) = \mathrm{constant} \times \int_{-\infty}^{+\infty}\int_{0}^{2\pi}\int_{0}^{1}\int_{-\infty}^{+\infty} dy\, d\varphi\, d\mu\, dx \times$$

$$10^{8} \times \left(0.036 \times \left(\sqrt{x^2+\rho^2}\right)^{-1.5} + 1.55 \times \left(\sqrt{x^2+\rho^2}\right)^{-6} + 2.99 \times \left(\sqrt{x^2+\rho^2}\right)^{-16} \right) \times$$

$$\frac{\mu}{\sqrt{x^2+\rho^2}\sqrt{x^2+\rho^2+\mu^2-1}} \times$$

$$\left(1-\left(1-\beta^2\right)\sin^2\varphi\right) \times$$

$$\frac{1-u_1(\xi)+\mu u_1(\xi)}{1-\frac{1}{3}u_1(\xi)} f(\xi) \times$$

$$\frac{\exp\left[-y^2\right]}{1+2b^2\beta\frac{w}{c}} \quad \text{where}$$

$$\mathrm{constant} = \frac{3\times\sigma_T (AU)^2}{16\pi^{5/2} R_{Solar}}$$

$$b = \frac{1}{\sqrt{2}}\left(1+\rho\sqrt{\frac{1-\beta^2}{x^2+\rho^2}}\sin\varphi + \frac{\beta\rho}{\sqrt{x^2+\rho^2}}\right)^{1/2}$$

$$\beta = \cos\left(\sin^{-1}\left(\sqrt{\frac{1-\mu^2}{x^2+\rho^2}}\right)\right)$$

$$\xi = \frac{\lambda\left(1+2\frac{q}{c}by\right)}{1+2b^2\beta\frac{w}{c}} \tag{B.34}$$

$$q = \sqrt{\frac{2kT}{m}}$$

$$\boxed{\begin{array}{l}\sigma_T = 6.677\times 10^{-25}\ \ \mathrm{cm}^2\\ R_{solar} = 6.96\times 10^{10}\ \ \mathrm{cm}\\ AU = 1.496\times 10^{13}\ \ \mathrm{cm}\end{array}} \tag{B.35}$$

From equation (**B.33**) and equation (**B.34**) the total observed intensity is given by equation (**B.36**).

$$I_{Total}(\lambda,\rho) = I_{//}^{Ra}(\lambda,\rho) + I_{\perp}^{Ra}(\lambda,\rho) \tag{B.36}$$

1. Evaluation of the $\int_{-\infty}^{+\infty} A(x)dx$ integral

The above integral is evaluated using a 20-point Hermite polynomial expansion as shown in equation (**B.37**).

$$\begin{aligned}\int_{-\infty}^{+\infty} A(x)dx &= \int_{-\infty}^{+\infty}\left(A(x)e^{x^2}\right)e^{-x^2}dx \\ &\cong \sum_{i=1}^{20} A(h_i)e^{(h_i)^2} w_i\end{aligned} \tag{B.37}$$

where h_i and w_i are the Hermite polynomials and the associated weights, respectively.

2. Evaluation of the $\int_0^{2\pi} \mathbf{B}(\varphi)\mathbf{d}\varphi$ integral

This integral is evaluated using 20-point trapezoidal composite rule as shown in equation (**B.38**).

$$\boxed{\begin{aligned}&\int_0^{2\pi} \mathbf{B}(\varphi)\mathbf{d}\varphi \cong \frac{\mathbf{h_B}}{\mathbf{2}}\left(2\times\sum_{j=0}^{20}\mathbf{B}(\varphi_j)-\mathbf{B}(\varphi_0)-\mathbf{B}(\varphi_{20})\right) \ \textbf{where}\\ &\mathbf{h_B} = \frac{\mathbf{2\pi-0}}{\mathbf{20}}\\ &\varphi_j = \mathbf{0}+\mathbf{h_B}\times\mathbf{j}\end{aligned}} \tag{B.38}$$

3. Evaluation of the $\int_0^{1} \mathbf{C}(\mu)\mathbf{d}\mu$ integral

This integral is evaluated using 10-point trapezoidal composite rule as shown in equation (**B.39**).

$$\boxed{\begin{aligned}&\int_0^{1} \mathbf{C}(\mu)\mathbf{d}\mu \cong \frac{\mathbf{h_C}}{\mathbf{2}}\left(2\times\sum_{k=0}^{10}\mathbf{C}(\mu_j)-\mathbf{C}(\mu_0)-\mathbf{C}(\mu_{10})\right) \ \textbf{where}\\ &\mathbf{h_C} = \frac{\mathbf{1-0}}{\mathbf{10}}\\ &\mu_k = \mathbf{0}+\mathbf{h_C}\times\mathbf{k}\end{aligned}} \tag{B.39}$$

4. Evaluation of the $\int_{-\infty}^{+\infty} \mathbf{D}(\mathbf{y})\mathbf{e}^{-y^2}\mathbf{dy}$ integral

This integral is evaluated using a 12-point Hermite polynomial expansion as shown in equation (**B.40**).

$$\int_{-\infty}^{+\infty} \mathbf{D}(\mathbf{y})\mathbf{e}^{-y^2}\mathbf{dy} \cong \sum_{l=1}^{12}\mathbf{D}(\mathbf{h}_l)\mathbf{w}_l \tag{B.40}$$

where $\mathbf{h}_l$ and $\mathbf{w}_l$ are the Hermite polynomials and the associated weights, respectively.

Consider the integral $\int_{-\infty}^{+\infty} \mathbf{D(y)e^{-y^2}\,dy}$ its **y** dependent parameters and the expansion outlines in equation (**B.40**) in equation (**B.41**).

$$\begin{aligned}
&\int_{-\infty}^{+\infty} \mathbf{D(y)e^{-y^2}\,dy} \equiv \int_{-\infty}^{+\infty} \frac{\mathbf{1-u_1(\xi)+\mu u_1(\xi)}}{\mathbf{1-\frac{1}{3}u_1(\xi)}}\mathbf{f(\xi)e^{-y^2}\,dy} \\
&\textbf{where } \xi = \xi(\mathbf{y}) \\
&\qquad \cong \sum_{l=1}^{12} \frac{\mathbf{1-u_1(\xi_l)+\mu u_1(\xi_l)}}{\mathbf{1-\frac{1}{3}u_1(\xi_l)}}\mathbf{f(\xi_l)w_l} \\
&\qquad = \sum_{l=1}^{12} \frac{\mathbf{1-u_1(\xi_l)}}{\mathbf{1-\frac{1}{3}u_1(\xi_l)}}\mathbf{f(\xi_l)w_l} + \mu\sum_{l=1}^{12} \frac{\mathbf{u_1(\xi_l)}}{\mathbf{1-\frac{1}{3}u_1(\xi_l)}}\mathbf{f(\xi_l)w_l} \\
&\textbf{where } \xi_l = \frac{\lambda\left(\mathbf{1+2\frac{q}{c}bh_l}\right)}{\mathbf{1+2b^2\beta\frac{w}{c}}}
\end{aligned} \qquad \textbf{(B.41)}$$

Due to the rapid intensity variations associated with the Fraunhofer lines, as evident in figure (**B.1**), this effect is included by taking the mean value for $\mathbf{f(\xi_l)}$ over each interval in the Hermite polynomial expansion coefficients $\mathbf{h_l}$. The mean is then given by equation (**B.42**).

$$\left\langle \frac{1-u_l(\xi_l)+\mu u_l(\xi_l)}{1-\frac{1}{3}u_l(\xi_l)} f(\xi_l) \right\rangle$$

$$= \frac{1}{\left(\frac{\xi_l+\xi_{l+1}}{2}\right)-\left(\frac{\xi_l+\xi_{l-1}}{2}\right)} \int_{\left(\frac{\xi_l+\xi_{l-1}}{2}\right)}^{\left(\frac{\xi_l+\xi_{l+1}}{2}\right)} \frac{1-u_l(\eta)+\mu u_l(\eta)}{1-\frac{1}{3}u_l(\eta)} f(\eta) d\eta$$

$$\equiv \frac{1}{\eta_R} \int_{\eta_{lower}}^{\eta_{upper}} \frac{1-u_l(\eta)+\mu u_l(\eta)}{1-\frac{1}{3}u_l(\eta)} f(\eta) d\eta$$

$$= \frac{1}{\eta_R} \int_{\eta_{lower}}^{\eta_{upper}} \frac{1-u_l(\eta)}{1-\frac{1}{3}u_l(\eta)} f(\eta) d\eta + \frac{\mu}{\eta_R} \int_{\eta_{lower}}^{\eta_{upper}} \frac{u_l(\eta)}{1-\frac{1}{3}u_l(\eta)} f(\eta) d\eta$$

$$\text{where } \eta_{upper} = \frac{\lambda - \lambda \frac{q}{c} b(h_l + h_{l+1})}{1+2b^2\beta \frac{w}{c}}$$

$$\eta_{lower} = \frac{\lambda - \lambda \frac{q}{c} b(h_l + h_{l-1})}{1+2b^2\beta \frac{w}{c}}$$

$$\eta_R = \eta_{upper} - \eta_{lower} \tag{B.42}$$

$$\text{for } l = 1,\ h_{l-1} = 0$$

$$\text{for } l = 12,\ h_{l+1} = 0$$

In order to evaluate the integral in equation (**B.42**) the following procedure was followed in order to expedite the calculation.

1. The extraterrestrial solar irradiance spectrum was recreated in intervals of 0.0025 Angstroms (η').

2. For a given set of values for $(\mathbf{x}, \boldsymbol{\varphi}, \boldsymbol{\mu})$ from the integrals $\left(\int \mathbf{dx}, \int \mathbf{d\varphi}, \int \mathbf{d\mu}\right)$ the values $(\eta_{upper}, \eta_{lower}, \eta_R)$ were calculated.

3. $(\eta_{upper}, \eta_{lower})$ were assigned new values $(\eta'_{upper}, \eta'_{lower})$ which are divisible by 0.0025 and closest of (η') to $(\eta_{upper}, \eta_{lower})$.

4. The values $(f(\eta_{upper}), f(\eta_{lower}))$ now corresponded to $(\eta'_{upper}, \eta'_{lower})$.

5. This procedure allowed the integral to be expanded in intervals of 0.0025 Angstroms.

6. Within the range of 0.0025 Angstroms the limb-darkening coefficient $\mathbf{u}_1$ extraterrestrial solar irradiance **f** could be assumed to be a constant. This allows for writing equation (**B.42**) as shown in equation (**B.43**).

$$
\boxed{
\begin{aligned}
&\frac{1}{\eta_R}\int_{\eta_{lower}}^{\eta_{upper}} \frac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta + \frac{\mu}{\eta_R}\int_{\eta_{lower}}^{\eta_{upper}} \frac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta \\
&\cong \frac{1}{\eta_R}\left(
\begin{aligned}
&\int_{\eta'_{lower}}^{\eta'_{lower}+0.0025A^0} \frac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta + \int_{\eta'_{lower}+0.0025A^0}^{\eta'_{lower}+0.0050A^0} \frac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta + \\
&\ldots\ldots.. + \int_{\eta'_{upper}-0.0025A^0}^{\eta'_{upper}} \frac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta
\end{aligned}
\right) \\
&+ \frac{\mu}{\eta_R}\left(
\begin{aligned}
&\int_{\eta'_{lower}}^{\eta'_{lower}+0.0025A^0} \frac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta + \int_{\eta'_{lower}+0.0025A^0}^{\eta'_{lower}+0.0050A^0} \frac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta + \\
&\ldots\ldots.. + \int_{\eta'_{upper}-0.0025A^0}^{\eta'_{upper}} \frac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)} f(\eta)d\eta
\end{aligned}
\right)
\end{aligned}
}
\tag{B.43}
$$

Now to elaborate on the method, by which the integration is expedited, consider the following table. Columns **1** and **4** are from section **B.6** and columns **2** and **3** are from section **B.7.** Here **T** is divisible by 0.00025.

Table (B.1). The dependence of the extraterrestrial solar irradiance and the limb-darkening functions with the wavelength.. The columns 1 and 4 are from section (B.6) and columns 2 and 3 are from section (B.7).

Wavelength η in Nanometers (nm)	$\dfrac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)}$	$\dfrac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)}$	Extraterrestrial Solar irradiance $f(\eta)$
T	A1	B1	C1
T+0.00025	A2	B2	C2
T+0.00050	A3	B3	C3
T+0.00075	A4	B4	C4
T+0.00100	A5	B5	C5
T+0.00125	A6	B6	C6
T+0.00150	A7	B7	C7
T+0.00175	A8	B8	C8
T+0.00200	A9	B9	C9
continued	continued	continued	continued

In the code T is 300.00025 nm and then increases in intervals of 0.00025 nm to 550.00000 nm.

Now create a new table with each row reflecting the cumulative sum of table (**B.1**) for the columns **2**, **3** and **4**.

Table (B.2). Cumulative values of the columns 2,3 and 4 of table (B.1).

T	A1*C1	B1*C1	C1
T+0.00025	A2*C2	B2*C2	C2
T+0.00050	A3*C3	B3*C3	C3
continued	continued	continued	continued

In the code the following format is used, where columns **2** and **3** are multiplied by column **4** of table (**B.2**).

Table (B.3). Multiplying column 4 of table (B.2) with column 2 and 3 of table (B.2). This is the format of the table used by the code.

T	A1*C1	B1*C1
T+0.00025	(A1*C1)+(A2*C2)	(B1*C1)+(B2*C2)
T+0.00050	(A1*C1)+(A2*C2)+ (A3*C3)	(B1*C1)+(B2*C2)+ (B3*C3)
T+0.00075	(A1*C1)+(A2*C2)+ (A3*C3)+ (A4*C4)	(B1*C1)+(B2*C2)+ (B3*C3)+ (B4*C4)
continued	continued	continued

Going back to equation **(B.43)** the integrals could be written as a summation, as shown in equation (**B.44**). Figure (**B.4**) illustrates the evaluation of the first integral of equation (**B.44**) where the integration under each interval is approximated as the area under the square. This procedure is justified on ground of the very small interval.

$$\frac{1}{\eta_R}\int_{\eta_{lower}}^{\eta_{upper}}\frac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)}f(\eta)d\eta+\frac{\mu}{\eta_R}\int_{\eta_{lower}}^{\eta_{upper}}\frac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)}f(\eta)d\eta$$
$$\cong\frac{\Delta\eta}{\eta_R}\sum_{\substack{\eta=\eta'_{lower}\\ \text{in intervals}\\ \text{of } 0.0025A^0}}^{\eta'_{upper}}\frac{1-u_1(\eta)}{1-\frac{1}{3}u_1(\eta)}f(\eta)+\frac{\mu\Delta\eta}{\eta_R}\sum_{\substack{\eta=\eta'_{lower}\\ \text{in intervals}\\ \text{of } 0.0025A^0}}^{\eta'_{upper}}\frac{u_1(\eta)}{1-\frac{1}{3}u_1(\eta)}f(\eta) \quad \textbf{(B.44)}$$
$$\text{where } \Delta\eta=0.0025A^0$$
$$\eta_R=\eta'_{upper}-\eta'_{lower}$$



Figure (B.4). Illustrating the integration procedure of the first integral in equation (B.44).

Taking the cumulative value of the parameter reflected on the y-axis of figure (**B.4**) would give the following, as shown in figure (**B.5**).



Figure (B.5). Plot of the cumulative y-axis values of figure (B.4) against the wavelength η'.

The same procedure as above could be performed for the second integral in equation (**B.44**) too.

In terms of expediting the calculation of the integral presented in equation (**B.42**), equation (**B.44**) allows the code do the calculation in two easy steps in conjunction with table (**B.3**) and illustrated by figure (**B.5**) for the first integral, as follows.

1. Find the corresponding row numbers, say, rows (**R,P**), respectively, which equal the values $\left(\eta'_{\mathbf{upper}}, \eta'_{\mathbf{lower}}\right)$ in column **1** of table (**B.3**) and as depicted in figure (**B.5**).

2. Find the values in columns **2** and **3** corresponding to the rows (**R,P**) in table (**B.3**) and as shown by the arrows cutting across the y-axis in figure (**B.5**).

3. Now calculate $\left[\left(\mathbf{2(R)-2(P)}\right)+\mu\left(\mathbf{3(R)-3(P)}\right)\right]\times\frac{\Delta\eta}{\eta_{\mathbf{R}}}$ where the numbers correspond to the column numbers. The first part $\mathbf{2(R)-2(P)}$ corresponds to the first integral in equation (**B.44**) and the second part $\mathbf{3(R)-3(P)}$ corresponds to the second integral in equation (**B.44**). In figure (**B.5**), $\frac{\left(\mathbf{2(R)-2(P)}\right)}{\eta_{\mathbf{R}}}$ evaluates the first integral in equation (**B.44**).

B.11 The symbols used in the code

The following symbols have been used in the code.

$$\boxed{\begin{array}{l} \mathbf{x} \rightarrow \mathbf{x} \\ \varphi \rightarrow \mathbf{y} \\ \mu \rightarrow \mathbf{z} \\ \eta_{\mathbf{upper}} \rightarrow \mathbf{cetaU} \\ \eta_{\mathbf{lower}} \rightarrow \mathbf{cetaL} \\ \eta'_{\mathbf{upper}} \rightarrow \mathbf{a(0,u)} \\ \eta'_{\mathbf{lower}} \rightarrow \mathbf{a(0,t)} \\ \mathbf{h_l} \rightarrow \mathbf{hermite\_12(l)} \\ \mathbf{w_l} \rightarrow \mathbf{weights\_12(l)} \\ \mathbf{h_i} \rightarrow \mathbf{hermite\_20(i)} \\ \mathbf{w_i} \rightarrow \mathbf{weights\_20(i)} \\ \mathbf{b} \rightarrow \mathbf{bb} \\ \rho \rightarrow \mathbf{rho} \\ \lambda \rightarrow \mathbf{lambda} \end{array}} \qquad \textbf{(B.45)}$$

In the code table (**B.3**) is given by **3000_5500_cum_irradiance.dat**

From equation (**B.2**) the following relationships could be derived.

$$\boxed{\begin{aligned} \text{angle\_1} &\equiv \sin(\omega) \\ &= \frac{\sin(\theta)}{\sqrt{x^2+\rho^2}} \\ &= \frac{\sin\left(\cos^{-1}(\cos\theta)\right)}{\sqrt{x^2+\rho^2}} \\ &= \frac{\sin\left(\cos^{-1}(\mu)\right)}{\sqrt{x^2+\rho^2}} \end{aligned}} \tag{B.46}$$

$$\boxed{\begin{aligned} \text{angle\_2} &\equiv \cos(w) \\ &= \cos\left(\sin^{-1}(\text{angle\_1})\right) \end{aligned}} \tag{B.47}$$

$$\boxed{\begin{aligned} \text{angle\_3} &\equiv \cos(\chi) \\ &= \frac{x}{\sqrt{x^2+\rho^2}} \end{aligned}} \tag{B.48}$$

$$\boxed{\begin{aligned} \text{angle\_4} &\equiv \sin(\chi) \\ &= \sin\left(\cos^{-1}(\text{angle\_3})\right) \end{aligned}} \tag{B.49}$$

$$\boxed{\begin{aligned} \text{angle\_5} &\equiv \cos(\pi-\Theta) \\ &= \cos(\omega)\cos(\chi)+\sin(\omega)\sin(\chi)\sin(\varphi) \\ &= \text{angle\_2}\times\text{angle\_3}+\text{angle\_1}\times\text{angle\_4}\times\sin(y) \end{aligned}} \tag{B.50}$$

$$\boxed{\begin{aligned} \text{angle\_6} &\equiv (\pi-\Theta) \\ &= \cos^{-1}(\text{angle\_5}) \end{aligned}} \tag{B.51}$$

$$\begin{aligned} bb &\equiv b \\ &= \cos(\gamma) \\ &= \cos\left(\frac{\pi - \Theta}{2}\right) \\ &= \cos\left(\frac{angle\_6}{2}\right) \end{aligned} \tag{B.52}$$

From equation (**B.16**) the following relationships could be derived.

$$\begin{aligned} cc\_t &\equiv \frac{16\pi}{3\sigma_T} Q_{\perp}^{Ra} \\ &= 1.0 - \sin^2(\varphi)\sin^2(\omega) \\ &= 1.0 - \sin^2(y)(angle\_1)^2 \end{aligned} \tag{B.53}$$

$$\begin{aligned} cc\_r &\equiv \frac{16\pi}{3\sigma_T} Q_{//}^{Ra} \\ &= 1.0 + \sin^2(\varphi)\sin^2(\omega) - 4\left(1 - b^2\right)b^2 \\ &= 1.0 + \sin^2(y)(angle\_1)^2 - 4\left(1 - (bb)^2\right)(bb)^2 \end{aligned} \tag{B.54}$$

B.12 The IDL code

The following is the code written in IDL where (;) is a comment symbol and ($) a command continuation symbol.

```
;README
;DECIDE ON THE RADIAL WIND VELOCITY IN KM\SEC
;DECIDE ON THE ISOTHERMAL CORONAL TEMPERATURE IN MK
;DECIDE ON THE VALUES FOR RHO IN SOLAR RADIUS
;000.0 0.50 1.10
;400.0 2.00 1.30
;800.0 1.50 1.40
;ENTER THE ABOVE INFORMATION IN BATCH_FILE.PRO

;COMPILE ECLIPSE_SIMULATION.PRO
;RUN     ECLIPSE_SIMULATION.PRO

;THE DATA WILL BE STORED AS
;'ES_w000.0_T0.50_r1.10.dat'
;'ES_w400.0_T2.00_r1.30.dat'
;'ES_w800.0_T1.50_r1.40.dat'



;CALULATING AVERAGE INTENSITY IN THE REGION [cetaU,cetaL]
function KURUCZ_TABLE,a,z,cetaU,cetaL

u=long(round((double(cetaU)-300.00025d0)/0.00025d0)) ;a(0,0)=300.00025
t=long(round((double(cetaL)-300.00025d0)/0.00025d0))  ;a(0,0)=300.00025
interval=a(0,u)-a(0,t)
irradiance=((a(1,u)-a(1,t))+z*(a(2,u)-a(2,t)))*0.00025d0/interva
;microwatts/cm squared/nm
return,irradiance ;microwatts/cm squared/nm
end

;CALCULATE FOURTH_INTEGRAL PARAMETRS AND THE VALUE
function calculate_fourth_integral_value ,a,lambda,rho,hermite_12,weights_12,x,y,z,w,T
q  =5508.0*sqrt(T)      ;thermal electron velocity in km/sec
c  =3.0d5                ;velocity of light in km/sec
final_sum_4=dblarr(2,1)
   angle_1=sin(acos(z))/sqrt(x^2+rho^2)        ;sin(omega)
   angle_2=cos(asin(angle_1))                  ;cos(omega)
```

```
        angle_3=x/sqrt(x^2+rho^2)              ;cos(chi)
        angle_4=sin(acos(angle_3))             ;sin(chi)

            angle_5=angle_2*angle_3 + angle_1*angle_4*sin(y)
            angle_6=acos(angle_5)

r=sqrt(x^2+rho^2)                                        ;radius
aa=1.0d8*(0.036/(r^1.5) + 1.55/(r^6) + 2.55/(r^16))      ;Baumbach model
bb=cos(angle_6/2.0)                                       ;cos(gamma)
cc_t=1.0-(sin(y)^2)*(angle_1^2)                           ;tangential
cc_r=1.0+(sin(y)^2)*(angle_1^2)-4.0*(1.0-bb^2)*(bb^2);radial
dd=(1.0*z)/(sqrt(x^2+rho^2)*sqrt(x^2+rho^2+z^2-1.0))
;changing d(cos(omega))to d(mu)
 ee=1.0+2.0*bb^2*angle_2*(w/c)                            ;wind component

N=11; Nth order hermite polynomial expansion is used
  sum_i_tangential=0.0
  sum_i_radial   =0.0

  for i=0,N do begin; first_loop

     if (i eq 0 ) then begin
       cetaL=(lambda-lambda*(q/c)*bb*hermite_12(i))/ee
       ;nanometers
         end else if (i ne 0) then begin
           cetaL=(lambda-lambda*(q/c)*bb*(hermite_12(i)+hermite_12(i-1)))/ee
           ;nanometers
     endif

     if (i eq 11) then begin
       cetaU=(lambda-lambda*(q/c)*bb*hermite_12(i))/ee
       ;nanometers
         end else if (i ne 11) then begin
           cetaU=(lambda-lambda*(q/c)*bb*(hermite_12(i)+hermite_12(i+1)))/ee
           ;nanometers
     endif

   if (((cetaU or cetaL) lt 300.00025d0) or ((cetaU or cetaL) gt 550.00000d0)) then begin
       final_sum_4(0,0)=999999999999.9999
       final_sum_4(1,0)=999999999999.9999
       goto,terminate_4
   endif
```

;CALCULATING THE AVERAGE INTENSITY CORRESPONDING TO ;[cetaU,cetaL]

```
          irradiance=KURUCZ_TABLE(a,z,cetaU,cetaL) ;microwatts/cm squared/nm
          FC=irradiance*1.0d-7                     ;W/cm squared/Angstroms


sum_i_tangential=sum_i_tangential+weights_12(i)*(FC)
sum_i_radial    =sum_i_radial    +weights_12(i)*(FC)

    endfor ;loop_i


final_sum_4(0,0)=sum_i_tangential*aa*cc_t*dd*(1.0/ee)
final_sum_4(1,0)=    sum_i_radial*aa*cc_r*dd*(1.0/ee)
terminate_4:
return,final_sum_4
end
```

;CALCULATE THIRD_INTEGRAL PARAMETRS AND THE VALUE

```
function calculate_third_integral_value ,a,lambda,rho,hermite_12,weights_12,x,y,w,T

;THIRD INTEGRAL BOUNDS
a3=0.0
b3=1.0
n3=10.0
h3=(b3-a3)/n3
f3=fix(n3)
sum_z_t=dblarr(f3+1)
sum_z_r=dblarr(f3+1)
total_sum_z_t=0.0
total_sum_z_r=0.0
final_sum_3=dblarr(2,1)

  for r=0,f3 do begin
  z=a3+r*h3        ;mu=cos(theta)
  final_sum_4=calculate_fourth_integral_value $
              (a,lambda,rho,hermite_12,weights_12,x,y,z,w,T)
    if ( final_sum_4(0,0) eq  999999999999.9999) then begin
    final_sum_3=final_sum_4
    goto, terminate_3
    endif
  sum_z_t(r)=final_sum_4(0,0)
  total_sum_z_t=total_sum_z_t+final_sum_4(0,0)
```

```
   sum_z_r(r)=final_sum_4(1,0)
   total_sum_z_r=total_sum_z_r+final_sum_4(1,0)
   endfor

final_sum_3(0,0)=(h3/2.0)*(2.0*total_sum_z_t-sum_z_t(0)-sum_z_t(n3))
final_sum_3(1,0)=(h3/2.0)*(2.0*total_sum_z_r-sum_z_r(0)-sum_z_r(n3))
terminate_3:
return, final_sum_3
end


;CALCULATE SECOND_INTEGRAL PARAMETERS AND THE VALUE
function calculate_second_integral_value ,a,lambda,rho,hermite_12,weights_12,x,w,T

;SECOND INTEGRAL BOUNDS
a2=0.0
b2=2.0*!pi
n2=20.0
h2=(b2-a2)/n2
f2=fix(n2)
sum_y_t=dblarr(f2+1)
sum_y_r=dblarr(f2+1)
total_sum_y_t=0.0
total_sum_y_r=0.0
final_sum_2=dblarr(2,1)

   for v=0,f2 do begin
   y=a2+v*h2 ; in radians
   final_sum_3=calculate_third_integral_value $
                    (a,lambda,rho,hermite_12,weights_12,x,y,w,T)
      if ( final_sum_3(0,0) eq  999999999999.9999) then begin
      final_sum_2=final_sum_3
      goto, terminate_2
      endif
   sum_y_t(v)=final_sum_3(0,0)
   total_sum_y_t=total_sum_y_t+final_sum_3(0,0)
   sum_y_r(v)=final_sum_3(1,0)
   total_sum_y_r=total_sum_y_r+final_sum_3(1,0)

   endfor

final_sum_2(0,0)=(h2/2.0)*(2.0*total_sum_y_t-sum_y_t(0)-sum_y_t(n2))
final_sum_2(1,0)=(h2/2.0)*(2.0*total_sum_y_r-sum_y_r(0)-sum_y_r(n2))
terminate_2:
```

```
return, final_sum_2
end
```

;CALCULATE FIRST INTEGRAL PARAMETERS AND THE VALUE

```
functioncalculate_first_integral_value $
,a,lambda,rho,hermite_12,weights_12,hermite_20,weights_20,w,T

;FIRST INTEGRAL USING HERMITE POLYNOMIALS
n1=19
sum_x_t=dblarr(n1+1)
total_sum_x_t=0.0
sum_x_r=dblarr(n1+1)
total_sum_x_r=0.0
intensity=dblarr(2,1)

   for h= 0,n1 do begin
   x=hermite_20(h)
   final_sum_2=calculate_second_integral_value $
                  (a,lambda,rho,hermite_12,weights_12,x,w,T)
      if (final_sum_2(0,0) eq 999999999999.9999) then begin
      intensity=final_sum_2
      goto, terminate_1
      endif
   sum_x_t(h)=final_sum_2(0,0)*weights_20(h)*exp(x*x)
   total_sum_x_t=total_sum_x_t+sum_x_t(h)
   sum_x_r(h)=final_sum_2(1,0)*weights_20(h)*exp(x*x)
   total_sum_x_r=total_sum_x_r+sum_x_r(h)
   endfor

sigma=6.677E-25          ;Thomson Scattering Cross Section cm squared
AU=1.49597870E+13     ;cm-symbol D in CRAM
solar_radius=6.96E+10   ;cm-symbol R in CRAM
constant=(3.0*sigma*(AU)^2.0)/(16.0*solar_radius*(!pi)^(2.5))

intensity(0,0)=constant*total_sum_x_t ;Watts/cm squared/Angstroms/steradians
intensity(1,0)=constant*total_sum_x_r ;Watts/cm squared/Angstroms/steradians

terminate_1:
return,intensity
end

print,'starting time ', systime(0)
```

;FILES TO BE READ
;READING THE KURUCZ EXTRATERRESTRIAL IRRADIANCE AT 0.00025A
;INTERVALS FROM 3000A TO 5500A

```
a=dblarr(3,1000000l)
openr,2,'3000_5500_cum_irradiance.dat'
readf,2,a
close,2
```

;READING THE ROOTS OF 12TH ORDER HERMITE POLYNOMIALS
;HANDBOOK OF TABLES OF MATHEMATICS:SELBY 3RD EDITION PAGE
;843

```
hermite_12=dblarr(1,12)
openr,3,'hermite_12_roots.dat'
readf,3,hermite_12
close,3
```

;READING THE WEIGHTS OF 12TH ORDER HERMITE POLYNOMIALS
;HANDBOOK OF TABLES OF MATHEMATICS:SELBY 3RD EDITION PAGE
;843

```
weights_12=dblarr(1,12)
openr,4,'hermite_12_weights.dat'
readf,4,weights_12
close,4
```

;READING THE ROOTS OF 20TH ORDER HERMITE POLYNOMIALS
;HANDBOOK OF MATHEMATIC FUNCTIONS EDITED BY M.ABRAMOWITZ
;AND L.A. STEGUN PAGE 924

```
hermite_20=dblarr(1,20)
openr,5,'hermite_20_roots.dat'
readf,5,hermite_20
close,5
```

```
;READING THE WEIGHTS OF 20TH ORDER HERMITE POLYNOMIALS
;HANDBOOK OF MATHEMATIC FUNCTIONS EDITED BY M.ABRAMOWITZ
;AND L.A. STEGUN PAGE 924

        weights_20=dblarr(1,20)
        openr,6,'hermite_20_weights.dat'
        readf,6,weights_20
        close,6


;READING THE BATCH_FILE.PRO
;THIS FILE CONTAINS THE WIND,TEMPERATURE AND RHO VALUES

data=read_ascii('batch_file.pro',$
comment_symbol='xxx.x',data_start=1)
data_elements=n_elements(data.field1)

counter=(data_elements/3)-1

for j=0,counter do begin

   wind=  data.field1(3*j+0);wind in km\sec
   temp=  data.field1(3*j+1);Temperature in MK
   rhoo=  data.field1(3*j+2);rho in solar radius

        str_w=string(wind)
        parts=str_sep(str_w,'.')
        w1=parts[0]
        w2=strmid(parts[1],0,1)

        str_t=string(temp)
        parts=str_sep(str_t,'.')
        t1=parts[0]
        t2=strmid(parts[1],0,2)

        str_r=string(rhoo)
        parts=str_sep(str_r,'.')
        r1=parts[0]
        r2=strmid(parts[1],0,2)

    w=double(wind)
    T=double(temp)
    rho=double(rhoo)
    print,'wind = ',w,' km\sec  temp = ',temp,' MK $
```

```
rho = ',rhoo,' solar radius'

;RECORDING THE OUTPUT

openw,9,'ES_w'+trim(w1,'(i3.3)')+'.'+trim(w2,'(i1.1)')+'_T'$
+trim(t1,'(i1.1)')+'.'+trim(t2,'(i2.2)')+'_r'+trim(r1,'(i1.1)')+'.'$
+trim(r2,'(i2.2)')+'.dat'

   for r=0,1000 do begin    ;370.0 nm to 470.0 nm
   lambda=370.0d0 + r*0.1d0 ;nano meters

   intensity=calculate_first_integral_value $
               (a,lambda,rho,hermite_12,weights_12,hermite_20,weights_20,w,T)
         if (intensity(0,0) eq 999999999999.9999) then begin
         print,'intensity(0,0) = 999999999999.9999 for $
                                    lambda = ',lambda
         goto, intensity_out_of_range
         endif

   total_intensity=intensity(0,0)+intensity(1,0);Watts/cm squared/Angstroms/steradians

   print,'lambda= ',lambda*10.0 ,'  Angstroms',' $
intensity = ',total_intensity*1.0E+7 ,$
        '  ergs/sec/cm squared/Angs/steradians'
   ;print,'time  = ',systime(0)
   printf, 9, format='(4E15.5)',lambda*10.0,total_intensity,intensity(0,0),$
intensity(1,0)
   flush, 9

   intensity_out_of_range:
   endfor ; the r loop for intensity

close, 9
endfor ; the j loop for batch_file
print,'ending time   ',systime(0)
end
```

An example of the input data information given in BATCHFILE.PRO

Table (B.4). The format of the input information on the wind velocity in km/sec, isothermal temperature in MK and the distance to the line of sight in SR.

Wind(km/sec)	Temperature(MK)	RHO(SR)
400.0	1.50	1.10
300.0	0.50	1.50
continued	continued	continued
XXX.X	XX.X	XX.X